ANALYTICAL BIOCHEMISTRY 24, 176–184 (1968)

Computer Analysis of Metal-Complexing Equilibria from pH Titration Data

H. STEWART HENDRICKSON

University of Texas Southwestern Medical School at Dallas, Departments of Biochemistry and Physiology, Dallas, Texas 75235

Received October 12, 1967

Metal ions are known to interact with a wide variety of biological compounds and are necessary cofactors in many enzymic reactions; thus, it is often necessary to know in quantitative terms the strength of these interactions. Also, in studying enzymic reactions of ionic compounds which are potential metal chelators, it is desirable to know the concentrations of the different ionic forms of substrate in solution at a given pH and metal ion concentration. This paper presents a general computer method for the determination of metal-ligand stability constants and mole fractions of ionic species in solution.

The calculation of metal-ligand stability constants has received considerable attention (1) since the classical work of Bjerrum (2) in 1941. Due to the complexity of the equilibria involved, these earlier methods have relied heavily on simplifying assumptions and graphic evaluations. With the advent of modern digital computers it is now practical to treat these problems in a more exact and rigorous manner. The usual method for evaluation of the stability of a metal complex of a polyprotic ligand involves a study of the competition between metal ion and hydrogen ion for the ligand. A pH titration curve of the ligand in the absence of complexing metal ions is first obtained in order to evaluate the acid association constants. The ligand is then titrated in the presence of a known amount of metal ion. The pH lowering of the titration curve in the presence of complexing metal ion is then a measure of the strength of metal binding.

CALCULATIONS

The calculation of stability constants involves an analysis of all the important equilibria involved in the titration. Assuming a 1:1 metal-ligand complex is formed, the following equilibria and equilibrium constants should be considered. Protonation of the ligand, H_NL, is described by the following general expression:

$$H^+ + H_{J-1}L^{-N+J-1} \rightleftharpoons H_JL^{-N+J}$$

$$K_{1J} = \frac{[H_JL^{-N+J}]}{[H^+][H_{J-1}L^{-N+J}]}$$

$$J = 1, 2, \ldots, N \tag{1}$$

For ease of notation, $K_{10} = 1$. Metal complexing by each species of ligand is described by the general expression:

$$M^{2+} + H_JL^{-N+J} \rightleftharpoons MH_JL^{-N+J+2}$$

$$K_{MH_JL} = \frac{[MH_JL^{-N+J+2}]}{[M^{2+}][H_JL^{-N+J}]}$$

$$J = 0, 1, 2, \ldots, N \tag{2}$$

Finally, the ionization of water has to be considered:

$$H_2O \rightleftharpoons H^+ + OH^- \qquad K_W = [H^+][OH^-] \tag{3}$$

Equations describing the total metal ion concentration (CM) and total ligand concentration (CL) can be written as follows.

$$CM = [M^{2+}] + \sum_{J=0}^{J=N} [MH_JL^{-N+J+2}] \tag{4}$$

$$CL = \sum_{J=0}^{J=N} [H_JL^{-N+J}] + \sum_{J=0}^{J=N} [MH_JL^{-N+J+2}] \tag{5}$$

Electroneutrality in solution requires that the total positive charges equal the total negative charges. This is shown in the following equation, where BA = moles OH^- added/mole ligand:

$$[H^+] + BA \cdot CL + 2[M^{2+}] = 2CM + [OH^-] + \sum_{J=0}^{J=N} \{(N - J)[H_JL^{-N+J}] + (N - J - 2)[MH_JL^{-N+J+2}]\} \tag{6}$$

The term BA · CL, on the left, represents the charge of the cation associated with the base titrant, and the term 2CM, on the right, represents the charge of the anion associated with the added metal ion. Substitution of equations 3 and 4 into equation 6 gives:

$$[H^+] + BA \cdot CL - \frac{K_W}{[H^+]} = \sum_{J=0}^{J=N} \{(N - J)\,[H_JL^{-N+J}] + (N - J)\,[MH_JL^{-N+J+2}]\} \tag{7}$$

The concentration of uncomplexed species of ligand is given by:

$$[H_JL^{-N+J}] = PK_J[H^+]^J[L^{-N}] \tag{8}$$

where

$$PK_J = \prod_{i=0}^{i=J} K_{1i} \tag{9}$$

Substitution of equation 8 into equation 7, and division by equation 5 gives:

$$\frac{[H^+]}{CL} + BA - \frac{K_W}{[H^+]CL} = \frac{\sum_{J=0}^{J=N} (N - J)PK_J[H^+]^J(1 + K_{MH_JL}[M^{2+}])}{\sum_{J=0}^{J=N} PK_J[H^+]^J(1 + K_{MH_JL}[M^{2+}])} \tag{10}$$

which, upon rearrangement, gives:

$$\sum_{J=0}^{J=N} (N - J - XA)PK_J[H^+]^J[M^{2+}]K_{MH_JL} = \sum_{J=0}^{J=N} (XA + J - N)PK_J[H^+]^J \tag{11}$$

where

$$XA = \frac{[H^+]}{CL} + BA - \frac{K_W}{[H^+]CL}$$

The following equation can be derived from equations 2, 4, 5, and 9:

$$\frac{CL}{CM - [M^{2+}]} = \frac{\sum_{J=0}^{J=N} PK_J[H^+]^J(1 + K_{MH_JL}[M^{2+}])}{\sum_{J=0}^{J=N} PK_J[H^+]^JK_{MH_JL}[M^{2+}]} \tag{12}$$

Equation 12 can be rearranged into the following quadratic form:

$$[M^{2+}]^2 \sum_{J=0}^{J=N} PK_J[H^+]^JK_{MH_JL} + [M^{2+}] \sum_{J=0}^{J=N} PK_J[H^+]^J(1 + CL \cdot K_{MH_JL} - CM \cdot K_{MH_JL}) - CM \sum_{J=0}^{J=N} PK_J[H^+]^J = 0 \tag{13}$$

This equation can be solved for $[M^{2+}]$ in the following form:

$$[M^{2+}] = \frac{-B + \sqrt{B^2 + 4AC}}{2A} \tag{14}$$

where

$$A = \sum_{J=0}^{J=N} PK_J[H^+]^J K_{MH_JL}$$

$$B = (CL - CM)A + \sum_{J=0}^{J=N} PK_J[H^+]^J$$

$$C = CM \sum_{J=0}^{J=N} PK_J[H^+]^J$$

For the titration of ligand in the absence of metal ion, equation 11 reduces to the following form:

$$\sum_{J=1}^{J=N} (N - J - XA)[H^+]^J PK_J = XA - N \qquad (15)$$

Equation 10 can be solved for BA in the following form:

$$BA = S4/S5 \qquad (16)$$

where

$$S4 = \sum_{J=0}^{J=N} (N - J - D)PK_J[H^+]^J(1 + K_{MH_JL}[M^{2+}])$$

$$S5 = \sum_{J=0}^{J=N} PK_J[H^+]^J(1 + K_{MH_JL}[M^{2+}])$$

$$D = \frac{[H^+]}{CL} - \frac{K_W}{[H^+]CL}$$

Acid association constants, K_{1J}, can be evaluated from the titration curve of the ligand in the absence of complexing metal ion. N sets of data (BA and $[H^+]$), from the buffering regions of the curve, are used to set up N simultaneous linear equations in the form of equation 15. These equations can then be solved for the N acid association products, PK_J, by the technique of matrix inversion (3) using a matrix inversion computer subroutine (4).

Once the acid association constants are known, stability constants can be calculated from a titration curve of ligand in the presence of metal ion. Sets of data (BA and $[H^+]$), from the buffering regions of the curve, along with first approximations of free metal ion concentrations can be used in equation 11 to set up N simultaneous linear equations, which are then solved for first approximations of stability constants, K_{MH_JL}. In order to solve N

equations for N unknowns, it is assumed that $K_{MH_NL} = 0$. Second approximations of free metal ion concentrations are obtained using equation 14. By an iterative process, exact values of the stability constants can be calculated.

Theoretical titration curves can be computed from the calculated acid association constants and stability constants using equation 16. The calculated constants can then be refined so that the best possible fit is obtained between the theoretical and experimental curves.

In the case of hydrolysis of the complex, ML, as described by the equilibrium expression:

$$ML + H_2O \rightleftharpoons MLOH + H^+ \qquad K_{MLOH} = \frac{[MLOH][H^+]}{[ML]}$$

equation 16 can be modified as follows:

$$BA = \frac{S4 + \frac{K_{MLOH}}{[H^+]}(N + 1 - D)K_{ML}[M^{2+}]}{S5 + \frac{K_{MLOH}}{[H^+]}K_{ML}[M^{2+}]} \tag{17}$$

An estimate of K_{MLOH} can be obtained from the titration curve and used in the calculation of these theoretical curves.

The mole fractions of ionic species H_JL and MH_JL are described by the following equations:

$$A_J = \frac{[H_JL^{-N+J}]}{CL}$$

$$B_J = \frac{[MH_JL^{-N+J+2}]}{CL}$$

By using equations 2, 5, and 8, these equations can be reduced to the following form:

$$A_J = \frac{PK_J[H^+]^J}{S5} \tag{18}$$

$$B_J = \frac{PK_J[H^+]^J K_{MH_JL}[M^{2+}]}{S5} \tag{19}$$

Equations 18 and 19 can then be solved for mole fractions of ionic species at a given pH and free metal ion concentration.

RESULTS AND DISCUSSION

The complete IBM 1620, Fortran II program for the calculation of stability constants is shown in Figure 1. In certain cases, especially when

```
C       PROGRAM TO CALCULATE STABILITY CONSTANTS (K(MH(J)L)) OF METAL
C       COMPLEXES OF H(N)L IN ALL OF ITS IONIC FORMS.
C
        DIMENSION A(10,10), XK(8), PH(8), BA(8), XM(8,32), XA(8), PK(8),
       1H(8), X(8,32), COMP(4)
        COMMON A
      1 READ 2, COMP, XO, XE, CM, CL, N
      2 FORMAT (4A4, F4.1, A2,2E10.3, I2)
        IF (XO)200,200,3
      3 PRINT 4, COMP, XO, XE, CL, CM, N
      4 FORMAT (///1X4A4, 3XF5.1,4H  X ,A2/6H CL = ,E10.3,5HCM = ,E10.3,5X
       14HN = ,I2)
        XN=N
        READ 5, (XK(I), I=1,N)
      5 FORMAT (8E10.3)
        XKW=1.0E-14
        M=N+1
        N1=N-1
        PRINT 6
      6 FORMAT(/4X2HBA,8X2HPH,8X3H(M),37X,10HITERATIONS)
      7 K=1
        DO 10 L=1,N
        READ 8, BA(L), PH(L), XM(L,K)
      8 FORMAT (2F10.2,E10.3)
        IF (BA(1)) 1,1,9
      9 H(L)=10.**(-PH(L))
     10 XA(L)= H(L)/CL + BA(L)-XKW/(H(L)*CL)
        PK(1)= XK(1)
        DO 15 L=2,N
     15 PK(L)= PK(L-1)*XK(L)
     16 DO 20 L=1,N
        A(L,N+1)=XA(L)-XN
        DO 17 I=1,N
        XI=I
     17 A(L,N+1)=(XA(L)+XI-XN)*PK(I)*H(L)**I + A(L,N+1)
        A(L,1)= (XN-XA(L))*XM(L,K)
        DO 20 J=1,N1
        XJ=J
     20 A(L,J+1)= (XN-XJ-XA(L))*PK(J)*XM(L,K)*H(L)**J
        CALL MATINV (A,N,M)
        DO 21 L=1,N
        IF (A(L,1))11,21,21
     11 A(L,1)=ABSF(A(L,1))
        PRINT 12
     12 FORMAT(/22H NEGATIVE VALUE A(L,1))
     21 X(L,K)=A(L,1)
        DO 30 L=1,N
        S2=0.
        S3=0.
        DO 22 J=1,N
     22 S3=S3+PK(J)*H(L)**J
        DO 23 J=1,N1
     23 S2=S2+PK(J)*X(J+1,K)*H(L)**J
        AA=X(1,K)+S2
```

FIG. 1A.

```
      BB=(CL-CM)*(X(1,K)+S2)+S3+1.
      CC=CM+CM*S3
      XM(L,K+1)=(-BB+SQRTF(BB**2+4.*AA*CC))/(2.*AA)
   30 CONTINUE
      DO 31 L=1,N
      DIFXM=ABSF(XM(L,K)-XM(L,K+1))
      IF(DIFXM - 0.0001*XM(L,K+1))31,31,36
   31 CONTINUE
      GO TO 40
   36 K=K+1
      IF (K-30)16,16,37
   37 PRINT 38
   38 FORMAT (/1X36HNO AGREEMENT AFTER 30 APPROXIMATIONS)
      GO TO 41
   40 K=K+1
   41 K1=K-1
      XLX=LOGF(X(1,K1))/2.303
      PRINT 43, BA(1),PH(1),XM(1,K),XLX,K1
   43 FORMAT (//2F10.3,E12.5,5X12HLOG K(ML) = ,F10.3,8XI3)
      DO 45 L=2,N
      XLX = LOGF(X(L,K1))/2.303
      L1=L-1
      PRINT 42, BA(L), PH(L), XM(L,K), L1, XLX
   42 FORMAT (/2F10.3,E12.5, 5X9HLOG K(MH(,I2,6H)L) = ,F10.3)
   45 CONTINUE
      GO TO 7
  200 CALL EXIT
      END
      SUBROUTINE MATINV (A,M,N)
C     FOR LISTING OF MATINV SEE REFERENCE(4).
C     INPUT DATA

 NTMP            1.0 CA   7.9E-04   7.9E-04 3
    7.8E 10  2.24E 07  8.36E 05
      0.488     5.700   7.0E-04
      1.595     6.900   3.5E-04
      2.47      8.900   1.8E-04
      0.0
                0.0
C
C     OUTPUT
C
 NTMP                1.0  X CA
CL =  7.900E-04CM =  7.900E-04     N =  3

   BA        PH         (M)                                        ITERATIONS

NEGATIVE VALUE A(L,1)

     .488     5.700 7.36878E-04     LOG K(ML) =      6.161          6

    1.595     6.900 3.89347E-04     LOG K(MH( 1)L) = 4.027

    2.470     8.900 1.60212E-04     LOG K(MH( 2)L) = 1.387
```

FIG. 1B.

FIGURE 1. Computer program.

$N > 3$, there is a tendency for round-off errors to occur in the matrix inversion subroutine. This is due to large differences in order of magnitude of the coefficients in the simultaneous linear equations. In this case it is advantageous to evaluate the titration in several small segments where $N \leqslant 3$. Round-off errors will often result in a negative value for a stability constant. Any negative values encountered are made positive in statement 11 and a "negative value" message is printed. This will often aid in convergence and eliminate further round-off errors. There is a safety valve built into the program which limits the calculations to 30 iterations.



FIG. 2. pH titration curves for nitrilotri(methylenephosphonic) acid: (A) free ligand, $7.9 \times 10^{-4} M$; (B) ligand + Ca(II), $7.9 \times 10^{-4} M$; Log K_1, log K_{12}, . . . , log K_{16}: 10.9, 7.35, 5.92, 4.60, 2.0, 1.9; log $K_{ML} = 6.25$, log $K_{MHL} = 4.15$, log $K_{MH_2L} = 2.7$; experimental points—open circles; theoretical curve—straight line.

Programs to calculate acid association constants and compute theoretical titration curves are quite straightforward and are not listed here. Listings of these programs can be obtained upon request to the author.

The use of these programs is illustrated by a study of the metal-binding properties of nitrilotri(methylenephosphonic) acid (NTMP) (5). The computer input and output listings for the determination of stability constants for the NTMP-Ca(II) complex are shown in Figure 1. Theoretical curves were computed for the titration of NTMP in the absence and presence of Ca(II) ion using the calculated and refined acid association constants and stability constants. As can be seen in Figure 2, good agreement was obtained between the theoretical and experimental curves. Table 1 shows the

TABLE 1
Mole Fractions of Ionic Species[a]
NTMP 1x Ca(II), CM = 7.9×10^{-4} *M*, CL = 7.9×10^{-4} *M*
pH = 7.0, BA = 4.654, $[Ca^{2+}] = 3.22 \times 10^{-4}$ *M*

L^{-6}	HL^{-5}	H_2L^{-4}	H_3L^{-3}	H_4L^{-2}
.000015	.118889	.266312	.022263	.000089
ML^{-4}	MHL^{-3}	MH_2L^{-2}	MH_3L^{-1}	
.008727	.540719	.042975	.000007	

[a] CM = total metal ion concentration, CL = total ligand concentration, BA = moles OH^-(titrant) added/mole ligand.

mole fractions of all the ionic species present at pH 7 in an equimolar solution of NTMP and Ca(II).

SUMMARY

Calculations are presented for the determination of acid association constants and metal-complex stability constants of polyprotic ligands from pH titration data. Equations are presented for the calculation of theoretical pH titration curves. Calculated constants can then be tested and refined by comparison of the experimental and theoretical curves. Finally, equations are given for the calculation of mole fractions of all ionic species in solution at a given pH and metal ion concentration.

ACKNOWLEDGMENT

This work was supported by a grant (GR 5443) from The National Science Foundation and a grant (H.E. 01574-13) from The National Institutes of Health of the U. S. Public Health Service.

REFERENCES

1. Rossotti, F. J. C., and Rossotti, H., "The Determination of Stability Constants." McGraw-Hill, New York, 1951.
2. Bjerrum, J., "Metal Ammine Formation in Aqueous Solution." P. Haase and Son, Copenhagen, 1941.
3. Searle, S. R., "Matrix Algebra for the Biological Sciences," p. 80. Wiley, New York, 1966.
4. Wiberg, K. B., "Computer Programing For Chemists," p. 44. Benjamin, New York, 1965.
5. Hendrickson, H. S., *Anal. Chem.* **39**, 998 (1967).